

ICWE 2012 Tutorial

An Introduction to SPARQL and Queries over Linked Data

...


Chapter 2: SPARQL

Olaf Hartig

<http://olafhartig.de/foaf.rdf#olaf>
@olafhartig


Database and Information Systems Research Group
Humboldt-Universität zu Berlin

SPARQL in General




- A family of W3C recommendations
- **SPARQL Query**
 - Declarative query language for RDF data
 - Our focus today
- **SPARQL Update**
 - Declarative update language for RDF data
- **SPARQL Protocol**
 - Communication between SPARQL processing services (a.k.a. SPARQL endpoints) and clients
- **SPARQL Query Results XML Format**
 - XML format for serializing query results

Main Idea of SPARQL Queries





- **Pattern matching:**
 - Describe subgraphs of the queried RDF graph
 - Subgraphs that match your description contribute an answer
 - Mean: **graph patterns** (i.e. RDF graphs with variables)





Main Idea of SPARQL Queries

Queried RDF graph:



Main Idea of SPARQL Queries


Queried RDF graph:



Result:


<code>?v</code>
<code>dbpedia:Mount_Baker</code> <code>dbpedia:Mount_Etna</code>

Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Prologue:**
 - **Prefix definitions** for using compact URIs (CURIEs)
 - Attention (difference to Turtle): **No period** (".") character as separator


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Result form specification:**
 - SELECT, DESCRIBE, CONSTRUCT, or ASK
(more about that later)


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Dataset specification:**
 - Specify the RDF dataset to be queried
(more about that later)

Components of SPARQL Queries




```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Query Pattern:**
 - WHERE clause specifies the graph pattern to be matched

- Different **types of graph patterns** for the query pattern (WHERE clause):
 - Basic graph pattern (BGP)
 - Group graph pattern
 - Optional graph pattern
 - Union graph pattern
 - Graph graph pattern
 - (Constraints)

Basic Graph Patterns



- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)
- Turtle syntax
 - Including syntactical sugar (e.g. property and object lists)


```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano .
    ?v rdfs:label ?name .
}
```

Basic Graph Patterns

- Set of triple patterns (i.e. RDF triples with variables)
- Variable names prefixed with “?” or “\$” (e.g. ?v, \$v)
- Turtle syntax
 - Including syntactical sugar (e.g. property and object lists)

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?name
WHERE {
    ?v rdf:type umbel-sc:Volcano ;
       rdfs:label ?name .
}
```

Basic Graph Patterns (Example)




```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano,  
                        umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data*

*Prefix definitions omitted

Basic Graph Patterns (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                        umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```


Data*

- Question: What are the names of all (known) volcanos?

```
SELECT ?name WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
    rdfs:label ?name . }  
Query*
```

*Prefix definitions omitted

Basic Graph Patterns (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                    umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data*

- Question: What are the names of all (known) volcanos?

```
SELECT ?name WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
    rdfs:label ?name . }
```


Query*

Result:

?name
"Etna"
"Беренберг"@ru
"Beerenberg"@en

*Prefix definitions omitted

Basic Graph Patterns (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                        umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: List all types of the volcano called “Beerenberg”


```
SELECT ?type WHERE {  
  ?v rdf:type ?type ;  
      rdfs:label "Beerenberg" .  
}
```

Query

?type

Empty!

Basic Graph Patterns (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano,  
                        umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: List all types of the volcano called “Beerenberg”


```
SELECT ?type WHERE {  
    ?v rdf:type ?type ;  
        rdfs:label "Beerenberg"@en .  
}
```

Query

?type

umbel-sc:Volcano
umbel-sc:NaturalElevation

Basic Graph Patterns (Example)




```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
                        p:location dbpedia:United_States .  
dbpedia:United_States rdfs:label "United States" .
```

Data

- **Question: Where are all (known) volcanos located?
(List the names of these locations)**

Basic Graph Patterns (Example)



```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
                      p:location dbpedia:United_States .  
dbpedia:United_States rdfs:label "United States" .
```

Data

- Question: Where are all (known) volcanos located?
(List the names of these locations)
- **Blank nodes** in SPARQL queries
 - As subject or object of a triple pattern
 - “Non-selectable” variables


```
SELECT ?name WHERE {  
  _:x rdf:type umbel-sc:Volcano ;  
      p:location [ rdfs:label ?name ] . }
```

Query

?name

"United States"

Basic Graph Patterns (Example)




```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
    p:location [ rdfs:label "United States"@en ,  
                  "États-Unis"@fr ] .  
  
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
    p:location [ rdfs:label "Italy" ] .
```

Data

- Blank nodes in the queried data

Basic Graph Patterns (Example)



```
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
    p:location [ rdfs:label "United States"@en ,  
                "États-Unis"@fr ] .  
  
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
    p:location [ rdfs:label "Italy" ] .
```

Data


- **Blank nodes in the queried data**
 - Blank node identifiers may occur in the results

```
SELECT ?l ?name WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
        p:location ?l .  
    ?l rdfs:label ?name .  
}
```

Query

?l	?name
_:x	"United States"@en
_:x	"États-Unis"@fr
_:y	"Italy"

Optional Graph Patterns



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en .
```

Data

- Question: What are **all** (known) volcanos and their names?


```
SELECT ?v ?name WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
  rdfs:label ?name . }
```

Query

- Problem: Mount Baker **missing** (it has no name)

?v	?name
dbpedia:Mount_Etna	"Etna"
dbpedia:Beerenberg	"Beerenberg"@en

Optional Graph Patterns



- Keyword **OPTIONAL** allows for optional patterns


```
SELECT ?v ?name WHERE {  
    ?v rdf:type umbel-sc:Volcano .  
    OPTIONAL { ?v rdfs:label ?name }  
}
```

Query

?v	?name
dbpedia:Mount_Etna	"Etna"
dbpedia:Mount_Baker	
dbpedia:Beerenberg	"Beerenberg"@en

- Optional patterns may result in unbound variables

Union Graph Patterns



Data


```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" ;  
                    p:location dbpedia:Italy .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
                    p:location dbpedia:United_States .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    p:location dbpedia:Norway .
```

- Question: What volcanos are located in Italy or in Norway?

Query

```
SELECT ?v WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
    p:location ? . } }
```

Union Graph Patterns




- Union graph patterns allow for alternatives

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Query

Union Graph Patterns




- Union graph patterns allow for alternatives

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Semantically
equivalent

```
SELECT ?v WHERE {  
  { ?v rdf:type umbel-sc:Volcano ;  
        p:location dbpedia:Italy }  
UNION  
  { ?v rdf:type umbel-sc:Volcano ;  
        p:location dbpedia:Norway }  
}
```

Group Graph Patterns



```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano .  
                  { ?v p:location dbpedia:Italy }  
                  UNION  
                  { ?v p:location dbpedia:Norway }  
}
```

Query


Semantically equivalent



```
SELECT ?v WHERE { { ?v rdf:type umbel-sc:Volcano }  
                  { { ?v p:location dbpedia:Italy }  
                    UNION  
                    { ?v p:location dbpedia:Norway } }  
}
```

Query

Constraints on Solutions




- **Syntax: Keyword FILTER followed by filter expression**

```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
PREFIX p: <http://dbpedia.org/property/>

SELECT ?v
WHERE {
    ?v rdf:type umbel-sc:Volcano ;
        p:lastEruption ?le .
    FILTER ( ?le > 1900 )
}
```

- **Filter expressions contain operators and functions**
- **Operators and functions operate on RDF terms**


Constraints (Truth Table)



- Filter expressions evaluate to true, false, or error
- Truth table:


A	B	A B	A && B
T	T	T	T
T	F	T	F
F	T	T	F
F	F	F	F
T	E	T	E
E	T	T	E
F	E	E	F
E	F	E	F
E	E	E	E

Unary Operators in Constraints



Operator	Type(A)	Result type
! A	xsd:boolean	xsd:boolean
+ A	numeric	numeric
- A	numeric	numeric
BOUND(A)	variable	xsd:boolean
isURI(A)	RDF term	xsd:boolean
isBLANK(A)	RDF term	xsd:boolean
isLITERAL(A)	RDF term	xsd:boolean
STR(A)	literal / URI	simple literal
LANG(A)	literal	simple literal
DATATYPE(A)	literal	simple literal

Constraints (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                           umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```


Data

- Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {  
  ?v rdf:type ?type ;  
      rdfs:label ?name .  
  FILTER ( STR(?name) = "Beerenberg" )  
}
```

Query

Constraints (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                           umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```


Data

- Question: List all types of the volcano called “Beerenberg”

```
SELECT ?type WHERE {  
  ?v rdf:type ?type ;  
     rdfs:label ?name .  
  FILTER ( STR(?name) = "Beerenberg" )  
}
```


Query	?type
	umbel-sc:Volcano
	umbel-sc:NaturalElevation

Constraints (Further Operators)



- Binary operators:
 - Logical connectives **&&** and **||** for **xsd:boolean**
 - Comparison operators **=**, **!=**, **<**, **>**, **<=**, and **>=** for numeric datatypes, **xsd:dateTime**, **xsd:string**, and **xsd:boolean**
 - Comparison operators **=** and **!=** for other datatypes
 - Arithmetic operators **+**, **-**, *****, and **/** for numeric datatypes
- Furthermore:
 - **REGEX**(String,Pattern) or **REGEX**(String,Pattern,Flags)
 - **sameTERM**(A,B)
 - **langMATCHES**(A,B)

Constraints (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                           umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: What volcanos have an “e” in their name?


```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
      rdfs:label ?name .  
  FILTER( REGEX(STR(?name),"e") )  
}
```

Query

?v

dbpedia:Beerenberg
dbpedia:Beerenberg

Constraints (Example)



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                           umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: What volcanos have an “e” in their name?


```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
      rdfs:label ?name .  
  FILTER( REGEX(STR(?name),"e","i") )  
}
```

Query

?v


dbpedia:Mount_Etna
dbpedia:Beerenberg
dbpedia:Beerenberg

Graph Graph Patterns



- SPARQL queries are executed over an **RDF dataset**:
 - One **default graph** and
 - Zero or more **named graphs** (identified by an URI)

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```


```
dbpedia:Mount_Baker http://example.org/d2
rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg http://example.org/d3
rdf:type umbel-sc:Volcano ;
rdfs:label "Beerenberg"@en .
```

- SPARQL queries are executed over an **RDF dataset**:

- One **default graph** and
- Zero or more **named graphs** (identified by an URI)


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Dataset specification:**
 - Specify the RDF dataset to be queried
(more about that later)


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Dataset specification:**
 - Specify the RDF dataset to be queried
~~(more about that later)~~ [here ...](#)
- **Specification using FROM and FROM NAMED**

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```


Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg http://example.org/d3  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Beerenberg"@en .
```

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph


```
dbpedia:Mount_Etna http://example.org/d1  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg http://example.org/d3  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Beerenberg"@en .
```

- Evaluation of patterns w.r.t. **active graph**
- **GRAPH** clause for making a named graph the active graph

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2
rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {
  GRAPH <http://example.org/d1> {
    ?v rdf:type umbel-sc:Volcano .
  }
}
```


```
http://example.org/d3
```

```
umbel-sc:Volcano ;
rdfs:label "Beerensberg"@en .
```

?v

dbpedia:Mount_Etna

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```


```
dbpedia:Mount_Baker http://example.org/d2
rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {
  GRAPH ?g {
    ?v rdf:type umbel-sc:Volcano .
  }
}
```

?v

```
dbpedia:Mount_Etna
dbpedia:Mount_Baker
dbpedia:Beerenberg
```

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2
rdf:type umbel-sc:Volcano .
```

```
SELECT ?v ?g WHERE {
```

```
  GRAPH ?g {
```


```
    ?v rdf:type
```

```
  }
```

```
}
```

?v	?g
dbpedia:Mount_Etna	<http://example.org/d1>
dbpedia:Mount_Baker	<http://example.org/d2>
dbpedia:Beerenberg	<http://example.org/d3>

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```


```
dbpedia:Mount_Baker http://example.org/d2
rdf:type umbel-sc:Volcano .
```

```
SELECT ?v WHERE {
  _:x rdfs:seeAlso ?g
  GRAPH ?g {
    ?v rdf:type umbel-sc:Volcano .
  }
}
```

?v

```
dbpedia:Mount_Etna
dbpedia:Mount_Baker
```

Negation




Data

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

- Question: What volcanos do **not** have a name in our data?

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: What volcanos do **not** have a name in our data?

```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  OPTIONAL { ?v rdfs:label ?name }  
  FILTER( ! BOUND(?name) )  
}
```


Query

?v

dbpedia:Mount_Baker

Negation as Failure

Negation




Data

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

- Question: What volcanos are **not** called “Beerenberg”?

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```


Data

- Question: What volcanos are **not** called “Beerenberg”?

```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
    rdfs:label ?name .  
  FILTER (STR(?name) != "Beerenberg")  
}
```

Query

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: What volcanos are **not** called “Beerenberg”?


```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  rdfs:label ?name .  
  FILTER (STR(?name) != "Beerenberg")  
}
```

Query

?v

dbpedia:Mount_Etna
dbpedia:Mount_Baker
dbpedia:Beerenberg

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Бере́нберг"@ru .
```

Data

- Question: What volcanos are **not** called "Beerenberg"?

```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  rdfs:label ?name .  
  FILTER (STR(?name) != "Beerenberg")  
}
```


Query

!=

?v

dbpedia:Mount_Etna
dbpedia:Mount_Baker
dbpedia:Beerenberg

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Бере́нберг"@ru .
```

Data

- Question: What volcanos are **not** called "Beerenberg"?

```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  rdfs:label ?name .  
  FILTER (STR(?name) != "Beerenberg")  
}
```


Query

!=

?v

dbpedia:Mount_Etna
dbpedia:Mount_Baker
dbpedia:Beerenberg

Negation



```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg  rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data

- Question: What volcanos are **not** called “Beerenberg”?


```
SELECT ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano .  
  OPTIONAL { ?v rdfs:label ?name .  
             FILTER (STR(?name) = "Beerenberg") }  
  FILTER ( ! BOUND(?name) )  
}
```

?v

dbpedia:Mount_Etna
dbpedia:Mount_Baker

Negation as Failure

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .  
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default
Graph


```
dbpedia:Mount_Etna http://example.org/d1  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Etna" .
```

```
dbpedia:Mount_Baker http://example.org/d2  
rdf:type umbel-sc:Volcano .
```

```
dbpedia:Beerenberg http://example.org/d3  
rdf:type umbel-sc:Volcano ;  
rdfs:label "Beerenberg"@en .
```

- **Question:** Which named graphs contain the name of a volcano that is not referenced in the default graph?

Graph Graph Patterns



```
dbpedia:Mount_Etna rdfs:seeAlso <http://example.org/d1> .
dbpedia:Mount_Baker rdfs:seeAlso <http://example.org/d2> .
```

Default

```
dbpedia:Mount_Etna http://example.org/d1
rdf:type umbel-sc:Volcano ;
rdfs:label "Etna" .
```


```
SELECT ?g WHERE {
  GRAPH ?g {
    ?v rdf:type umbel-sc:Volcano ;
    rdfs:label ?name .
  }
  OPTIONAL { ?v rdfs:seeAlso ?r }
  FILTER ( ! BOUND(?r) )
}
```

```
http://example.org/d2
sc:Volcano .
```

```
http://example.org/d3
-sc:Volcano ;
erenberg"@en .
```


ain the name of a
ne default graph?

Summary – Graph Patterns



- Different **types of graph patterns** for the query pattern (WHERE clause):
 - Basic graph pattern (BGP)
 - Group graph pattern
 - Optional graph pattern – keyword OPTIONAL
 - Union graph pattern – keyword UNION
 - Graph graph pattern – keyword GRAPH
 - Constraints – keyword FILTER


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Result form specification:**
 - SELECT, DESCRIBE, CONSTRUCT, or ASK
(more about that later)

Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Result form specification:**
 - SELECT, DESCRIBE, CONSTRUCT, or ASK
 - ~~(more about that later)~~ [here ...](#)

- **SELECT**

- Result: sequence of solutions (i.e. sets of **variable bindings**)
- Selected variables separated by space (not by comma!)
- Asterisk character (“*”) selects all variables in the pattern

- **SELECT**

- Result: sequence of solutions (i.e. sets of **variable bindings**)
- Selected variables separated by space (not by comma!)
- Asterisk character (“*”) selects all variables in the pattern

- **ASK**

- Check whether there is **at least one result**
- Result: true or false
- Example: Do we have data about volcanos?

```
ASK WHERE {  
    ?v rdf:type umbel-sc:Volcano .  
}
```

Query

- **DESCRIBE**

- Result: an RDF graph with data about resources
- Non-deterministic (i.e. query processor defines the actual structure of the resulting RDF graph)
- Example: just name the resource

```
DESCRIBE <http://dbpedia.org/resource/Beerenberg>
```

Query

Result Forms



- **DESCRIBE**

- Result: an RDF graph with data about resources
- Non-deterministic (i.e. query processor defines the actual structure of the resulting RDF graph)
- Example: just name the resource

```
DESCRIBE <http://dbpedia.org/resource/Beerenberg>
```

Query

- Example: Specify the resource(s) with a query pattern

```
DESCRIBE ?v WHERE {  
  ?v rdf:type umbel-sc:Volcano ;  
  rdfs:label ?name .  
}
```

Query

- Multiple variables possible or asterisk (“*”) for all

- **CONSTRUCT**

- Result: an RDF graph constructed from a template
- Template: graph pattern with variables from the query pattern

- **CONSTRUCT**

- Result: an RDF graph constructed from a template
- Template: graph pattern with variables from the query pattern

```
CONSTRUCT { ?v rdfs:label ?name ;  
             rdf:type myTypes:VolcanosOutsideTheUS  
}  
WHERE {  
    ?v rdf:type umbel-sc:Volcano ;  
       rdfs:label ?name .  
    OPTIONAL { ?v p:location ?l  
               FILTER ( ?l = dbpedia:United_States ) }  
    FILTER ( ! BOUND(?l) )  
}
```

Query

Result Forms




Data

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" ;  
                    p:location dbpedia:Italy .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Mount Baker" ;  
                    p:location dbpedia:United_States .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Beerenberg"@en ;  
                    p:location dbpedia:Norway .
```

Result

```
dbpedia:Mount_Etna rdfs:label "Etna" ;  
                    rdf:type myTypes:VolcanosOutsideTheUS .  
dbpedia:Beerenberg rdf:type myTypes:VolcanosOutsideTheUS ;  
                    rdfs:label "Beerenberg"@en .
```


Components of SPARQL Queries



```
PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
PREFIX umbel-sc: <http://umbel.org/umbel/sc/>
SELECT ?v
FROM <http://example.org/myGeoData>
WHERE {
    ?v rdf:type umbel-sc:Volcano .
}
ORDER BY ?name
```

- **Solution modifiers:**
 - Only for SELECT queries
 - Modify the **result set** as a whole (not single solutions)
 - Keywords: DISTINCT, ORDER BY, LIMIT, and OFFSET

Solution Modifiers



- **DISTINCT** removes duplicates from the result set

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano.  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                    umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Бере́нберг"@ru .
```

Data


```
SELECT ?type  
WHERE { _:x rdf:type ?type }
```

Query

?type

```
umbel-sc:Volcano  
umbel-sc:Volcano  
umbel-sc:NaturalElevation  
umbel-sc:Volcano
```

Solution Modifiers



- **DISTINCT** removes duplicates from the result set

```
dbpedia:Mount_Etna rdf:type umbel-sc:Volcano ;  
                    rdfs:label "Etna" .  
dbpedia:Mount_Baker rdf:type umbel-sc:Volcano .  
dbpedia:Beerenberg rdf:type umbel-sc:Volcano,  
                    umbel-sc:NaturalElevation ;  
                    rdfs:label "Beerenberg"@en ;  
                    rdfs:label "Беренберг"@ru .
```

Data


```
SELECT DISTINCT ?type  
WHERE { _:x rdf:type ?type }
```

Query

?type

```
umbel-sc:Volcano  
umbel-sc:NaturalElevation
```

Solution Modifiers




- **ORDER BY** orders the results

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano ;  
                  rdfs:label ?name }
```

Query

```
ORDER BY ?name
```

Solution Modifiers



- **ORDER BY** orders the results

```
SELECT ?v WHERE { ?v rdf:type umbel-sc:Volcano ;  
                  rdfs:label ?name }
```

Query

```
ORDER BY ?name
```


- How do we order different kinds of elements?
unbound variable < blank node < URI < literal
- **ASC** for ascending (default) and **DESC** for descending
- Hierarchical order criteria:

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ;  
                       p:lastEruption ?le ;  
                       rdfs:label ?name }
```

Query

```
ORDER BY DESC(?le), ?name
```

Solution Modifiers




- **LIMIT** – limits the number of results

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query
                        rdfs:label ?name }

ORDER BY ?name

LIMIT 5
```


Solution Modifiers



- **LIMIT** – limits the number of results

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query  
                      rdfs:label ?name }  
  
ORDER BY ?name  
LIMIT 5
```


- **OFFSET** – position/index of the first reported results

```
SELECT ?name WHERE { ?v rdf:type umbel-sc:Volcano ; Query  
                      rdfs:label ?name }  
  
ORDER BY ?name  
LIMIT 5 OFFSET 10
```

- Order of result should be predictable
(i.e. combine with **ORDER BY**)

- **New features of SPARQL 1.1 Query:**
 - Aggregate functions (e.g. COUNT, SUM, AVG)
 - Subqueries
 - Negation (EXISTS, NOT EXISTS, MINUS)
 - Assignments (e.g. BIND, SELECT expressions)
 - Property paths
 - Basic query federation (SERVICE, BINDINGS)
- **SPARQL 1.1 Update:**
 - Graph update (INSERT DATA, DELETE DATA, INSERT, DELETE, DELETE WHERE, LOAD, CLEAR)
 - Graph management (CREATE, DROP, COPY, MOVE, ADD)

Outline



Chapter 1:
Linked Data and

Chapter 2:
The SPARQL Query Language

Chapter 3:
Querying Linked Data on the Web

Now: Hands-on



<http://olafhartig.de/icwe.html>

**These slides have been created by
Olaf Hartig**

<http://olafhartig.de>

**This work is licensed under a
Creative Commons Attribution-Share Alike 3.0 License**
(<http://creativecommons.org/licenses/by-sa/3.0/>)

